Example 0.1 *a new uncharted planet has been detected, which has the typical circular ellipsoid shape of a planet which rotates about a central axis. The distance between the North and South Poles (the central axis length) is 2800 kilometers, and the planet's equator is circular with a diameter of 3600 kilometers. Find the volume of the atmosphere, which is 30 kilometers deep.*

The equator, by definition, lies in a plane which is the perpendicular bisector of the line segment between the North and South poles. The assignment of the north and south directions to the poles naturally is arbitrary.

Place the origin of an x, y, z-coordinate system at the center of the planet, with the North Pole on the positive z-axis, and the South Pole on the negative z-axis. The planet's surface is then defined by the equation

$$\frac{x^2}{1800^2} + \frac{y^2}{1800^2} + \frac{z^2}{1400^2} = 1.$$

The volume of the atmosphere is more difficult to compute than it may appear to be at first glance. The ellipsoid is almost spherical. If it were, then the variable ρ In the spherical coordinate system could have been used to specify the altitude of a point above the planet's surface. As it stands, radial lines through the origin are not quite orthogonal to the surface. Consequently, if $(\rho_0, \theta_0, \phi_0)$ represents a point on the surface of the planet, then $(\rho_0 + 30, \theta_0, \phi_0)$ represents a point in the atmosphere which is (except for a few special points) not quite 30 kilometers in altitude. This complication ultimately dooms the problem, and prevents a direct decimal solution.

Setting up a triple integral with ρ ranging from a point on the planet's surface, say $\rho = \rho_0$ to $\rho = \rho_0 + 30$ would appear to be a reasonable way to approximate the volume of the atmosphere, but it can be shown that it is not very accurate. The approach we use is both simpler and more accurate. Equally important, we are able to demonstrate the accuracy of the models used to approximate the volume.

The equation of the surface representing the upper limits of the atmosphere is very complicated, and a simpler equation will be used instead. It would seem reasonable to simply add 30 to the elliptical dimensions of 1800 and 1400 which define the equation of the planet's surface. The resulting equation

$$\frac{x^2}{1830^2} + \frac{y^2}{1830^2} + \frac{z^2}{1430^2} = 1$$

turns out to be a reasonably good approximation of the boundary of the atmosphere.

With such a simple model, an approximate volume of the atmosphere will be easy to compute. Most of the effort in the Maple work session below will be devoted to measuring the accuracy of the model. We will show that the approximation is an under approximation. By adding slightly more than 30 to the dimensions of the ellipsoid, which define the planet's surface, we will form yet another good approximation of the upper boundary of the atmosphere. Using this model leads to an over approximation for the volume of the atmosphere. The actual volume, then, will be in between these two reasonably close approximations.

****

```
> planet:=(x^2+y^2)/1800^2+z^2/1400^2=1;
```

$$planet := \frac{1}{3240000}x^2 + \frac{1}{3240000}y^2 + \frac{1}{1960000}z^2 = 1$$

```
> top:=(x^2+y^2)/1830^2+z^2/1430^2=1;
```

$$top := \frac{1}{3348900}x^2 + \frac{1}{3348900}y^2 + \frac{1}{2044900}z^2 = 1$$

#We plan to show that every point 30 kilometers above the planet's surface is outside of (or on) *top*. This will establish our claim that the volume of the region between *planet*, and *top* is an under approximation for the volume of the atmosphere.

Because of the rotational symmetry of the surfaces, this matter can be addressed in a simplified setting. We consider instead the ellipse generated by setting $y = 0$ in each of the two equations.#

```
> planet0:=subs(y=0,planet);
> top0:=subs(y=0,top);
```

$$planet0 := \frac{1}{3240000}x^2 + \frac{1}{1960000}z^2 = 1$$

$$top0 := \frac{1}{3348900}x^2 + \frac{1}{2044900}z^2 = 1$$

#Let $\vec{r}$ be a vector from the origin to a point (x, z) on the ellipse *planet*0, and let $r\vec{n}$ be a vector of length 30, which is normal to the ellipse *planet*0 at (x, z). We show that the components of the vector $\vec{r} + r\vec{n}$ (that is, the coordinates of the head of this vector) are always outside of *top*0, the second ellipse. The figure shown below might help, although nothing is drawn to scale.



It is convenient to produce the vectors $\vec{r}$, and $r\vec{n}$ parametrically.#

```
> with(linalg):
```

```
Warning, new definition for norm

Warning, new definition for trace
> r:=vector([1800*cos(t),1400*sin(t)]);
```
> #This is the position vector for the ellipse *planet*0.#

$$r := [1800\cos(t),\ 1400\sin(t)]$$

```
> rp:=map(diff,r,t);
```
> #The vector $\vec{rp}$ is tangent to the ellipse.#

$$rp := [-1800\sin(t),\ 1400\cos(t)]$$

#To create a normal vector $\vec{rn}$, simply switch the components of $\vec{rp}$, and change the sign of one of its components to create a vector whose dot product with $\vec{rp}$ is zero. It is a simple matter to decide which component should take the change in sign in order to produce an outer normal rather than an inner normal.#

```
> n:=vector([rp[2],-rp[1]]);
```

$$n := [1400\cos(t),\ 1800\sin(t)]$$

```
> r30:=matadd(r,scalarmul(n,30/sqrt(n[1]^2+n[2]^2)));
```

$$r30 := \left[1800\cos(t) + 210\,\frac{\cos(t)}{\sqrt{49\cos(t)^2 + 81\sin(t)^2}},\right.$$
$$\left. 1400\sin(t) + 270\,\frac{\sin(t)}{\sqrt{49\cos(t)^2 + 81\sin(t)^2}}\right]$$

```
> f:=subs(x=r30[1],z=r30[2],lhs(top0));
```

$$f := \frac{1}{3348900}\left(1800\cos(t) + 210\,\frac{\cos(t)}{\sqrt{49\cos(t)^2 + 81\sin(t)^2}}\right)^2$$
$$+ \frac{1}{2044900}\left(1400\sin(t) + 270\,\frac{\sin(t)}{\sqrt{49\cos(t)^2 + 81\sin(t)^2}}\right)^2$$

To show that the point corresponding to the head of the vector $\vec{r30}$ is always outside of (or on) the ellipse *top*0, it is enough to show that $f \geq 1$ for all t. Actually, because of symmetry, it is enough to show this for all t between 0 and $\pi/2$.#

```
> plot(f,t=0..Pi/2);
```

#This establishes our claim that the volume of the region between *planet*, and *top* is less than the volume of the atmosphere.



Before we compute the volume, we first create a slightly larger ellipse that will contain the entire atmosphere in its interior. This will lead to an over approximation for the volume of the atmosphere.#

```
> Top:=(x^2+y^2)/(1830+1/3)^2+z^2/(1430+1/3)^2=1;
```

$$Top := \frac{9}{30151081}\,x^2 + \frac{9}{30151081}\,y^2 + \frac{9}{18412681}\,z^2 = 1$$

```
> Top0:=subs(y=0,Top);
```

$$Top0 := \frac{9}{30151081}\,x^2 + \frac{9}{18412681}\,z^2 = 1$$

```
> F:=subs(x=r30[1],z=r30[2],lhs(Top0));
```

$$F := \frac{9}{30151081}\,(1800\cos(t) + 210\,\frac{\cos(t)}{\sqrt{49\cos(t)^2+81\sin(t)^2}})^2 + \frac{9}{18412681}\,(1400\sin(t) + 270\,\frac{\sin(t)}{\sqrt{49\cos(t)^2+81\sin(t)^2}})^2$$

#To show that the point corresponding to the head of the vector $\vec{r30}$ is always inside of (or on) the ellipse $Top0$, it is enough to show that $f \le 1$ for all t. Again, because of symmetry, it is enough to show this for all t between 0 and $\pi/2$.#

```
> plot(F,t=0..Pi/2);
```

#The volume of the region between *planet*, and *Top* will therefore be greater than the volume of the atmosphere.

All that is left to do is evaluate the two volumes. We use spherical coordinates to set up the integrals.#

```
> planet:=subs(x^2=rho^2-y^2-z^2,z=rho*cos(phi),planet);
```



$$planet := \frac{1}{3240000}\,\rho^2 + \frac{1}{4961250}\,\rho^2\cos(\phi)^2 = 1$$

```
> top:=subs(x^2=rho^2-y^2-z^2,z=rho*cos(phi),top);
```

$$top := \frac{1}{3348900}\,\rho^2 + \frac{652}{3424082805}\,\rho^2\cos(\phi)^2 = 1$$

```
> Top:=subs(x^2=rho^2-y^2-z^2,z=rho*cos(phi),Top);
```

$$Top := \frac{9}{30151081}\,\rho^2 + \frac{105645600}{555162236258161}\,\rho^2\cos(\phi)^2 = 1$$

```
> Rplanet:=solve(planet,rho);
```

$$Rplanet := \frac{12600}{\sqrt{49+32\cos(\phi)^2}},\ -\frac{12600}{\sqrt{49+32\cos(\phi)^2}}$$

```
> Rplanet:=Rplanet[1];
```

$$Rplanet := \frac{12600}{\sqrt{49+32\cos(\phi)^2}}$$

```
> Rtop:=solve(top,rho);
```

$$Rtop := \frac{261690}{\sqrt{20449+13040\cos(\phi)^2}},\ -\frac{261690}{\sqrt{20449+13040\cos(\phi)^2}}$$

```
> Rtop:=Rtop[1];
```

$$Rtop := \frac{261690}{\sqrt{20449 + 13040\cos(\phi)^2}}$$

```
> RTop:=solve(Top,rho);
```

$$RTop := 70685643\,\frac{\sqrt{18412681 + 11738400\cos(\phi)^2}}{165714129 + 105645600\cos(\phi)^2},\\ -70685643\,\frac{\sqrt{18412681 + 11738400\cos(\phi)^2}}{165714129 + 105645600\cos(\phi)^2}$$

```
> RTop:=simplify(RTop[1]);
```

$$RTop := \frac{23561881}{3}\,\frac{1}{\sqrt{18412681 + 11738400\cos(\phi)^2}}$$

```
> vol:=evalf(2*Pi)*evalf(int(int(rho^2*sin(phi),
> rho=Rplanet..Rtop),phi=0..Pi));
```

$$vol := .1059458140\,10^{10}$$

```
> Vol:=evalf(2*Pi)*evalf(int(int(rho^2*sin(phi),
> rho=Rplanet..RTop),phi=0..Pi));
```

$$Vol := .1071444219\,10^{10}$$

****

The volume of the atmosphere is between vol, and Vol. On the negative side, it should be said that an approximation technique is of greater value if it can be improved, and the technique used in this problem does not appear to offer room for improvement.